

## Maple Lecture 25. Three dimensional plots

In three dimensions we can plot surfaces and space curves. Just as in the previous lecture, we start loading the plots package:

```
[> with(plots);
```

We cover some of [3, Chapter 15], see [1], [2], and [4] for more examples.

### 25.1 Explicit and Implicit Plots

Surfaces can be defined as  $z = f(x, y)$ .

```
[> f := (x,y) -> cos(x*y):
[> plot3d(f, -2*Pi..2*Pi, -2*Pi..2*Pi, axes=box);
```

If we click on the plot, we adjust the orientation and view the object from any angle.

Many surfaces are defined implicitly. For example, the Whitney umbrella is defined by the equation  $x^2 - zy^2 = 0$ .

```
[> umbfun := (x,y,z) -> x^2 - z*y^2:
[> umbrella := implicitplot3d(umbfun, -2..2, -2..2, -2..2, grid=[20,10,10]):
[> display(umbrella);
```

The problem with this plot is that the handle of the umbrella is missing. From the equation we see that  $x = 0, y = 0$  is a solution for any value of  $z$ . But, we do not see the  $z$ -axis on the plot.

```
[> handle := plottools[line]([0,0,-2],[0,0,2]):
[> display(whitney,handle);
```

With this patch, we see that the umbrella is really not drawn very well... Maple does not consider the algebraic structure when plotting.

### 25.2 Options of plot3d

We can arrange the orientation to create a spin plot.

```
[> f := (x,y) -> x^3 - 3*x*y^2:
[> plotargs := f, -1..1, -1..1, axes=box:
[> plot3d(plotargs);
```

The orientation is described by two angles in spherical coordinates.

```
[> plot3d(plotargs, orientation=[45,45]);
```

As you can see, [45,45] is the default orientation. The first angle is the angle with the  $x$ -axis, we change this angle to turn the object to the left or right. The second angle is the angle with the  $z$ -axis. To see the object from above or below we change this axis.

```
[> spin := [seq(plot3d(plotargs, orientation=[10*i,45]), i=0..36)]:
[> display(spin, insequence=true);
```

As with two dimensional plots, the plotting happens in two stages: first the plot is computed and then rendered. This allows to make lists and sequences of plots, followed by the invocation of the display command.

## 25.3 Plotting Curves as Tubes

With `tubeplot` we make a tube around a space curve. Let us make a tubeplot from a circle.

```
[> x := cos(t): y := sin(t): z := 0:
[> torus1 := tubeplot([x,y,z],t=0..2*Pi,radius=1/4):
[> display(torus1);
[> torus2 := tubeplot([x+0.8,z,y],t=0..2*Pi,radius=1/4):
[> display([torus1,torus2]);
```

Let us make a chain of 10 tori :

```
[> chain := []:
[> optstube := t=0..2*Pi,radius=1/4:
[> dx := 0:
[> for i from 1 to 10 do
[>   if i mod 2 = 0
[>     then torus := tubeplot([2*x+dx,2*y,z],optstube):
[>     else torus := tubeplot([2*x+dx,z,2*y],optstube):
[>   end if:
[>   chain := [op(chain),torus]:
[>   dx := dx + 3:
[> end do:
[> display(chain,scaling=constrained);
```

## 25.4 Data plotting

We can make histograms from matrices:

```
[> indfun := (i,j) -> max(sin(i)*cos(j),0);
[> city := matrix(20,20,indfun):
[> matrixplot(city,heights=histogram);
```

## 25.5 Animation

Here we will make an animation of a torus knot:

```
[> r := 2 + 4/5*cos(7*t): z := sin(7*t):
[> curve := [r*cos(4*t), r*sin(4*t), z]:
[> picts := [seq(tubeplot(curve,t=0..2*Pi*i/20,radius=1/4),i=1..20)]:
[> display(picts,insequence=true,style=patch);
```

## 25.6 Riemann Surfaces

The logarithm for complex arguments defines a four dimensional surface: we let the real and imaginary part of the argument correspond to the usual  $x$  and  $y$  axes of three space. The vertical  $z$ -axis represents the real part of the function value, while the coloring of the surface is a function of the imaginary part.

```
[> w := u + I*v;
[> z := evalc(exp(w));
```

Observe how `evalc` expresses  $z$  as a function of  $u$  and  $v$ . As  $z$  equals  $\exp(w)$ ,  $w$  equals the natural logarithm of  $z$ . So we take  $x$  and  $y$  to be real and imaginary parts of  $z$ :

```
[> x := evalc(Re(z));
[> y := evalc(Im(z));
```

Then we plot the values for the complex logarithm in two ways, depending on what we choose as height and coloring.

First take the height to be the real value  $u$  of  $w$  and then take the coloring as the imaginary value  $v$  of  $w$ :

```
[> plot3d([x,y,u],u=-4..1,v=-3*Pi..3*Pi,orientation=[-56,72],colour=v);
```

Alternatively, we take height to be  $v$  and color with  $u$ :

```
[> plot3d([x,y,v],u=-4..1,v=-3*Pi..3*Pi,orientation=[-56,72],colour=u);
```

For more about complex plots, we refer to [1].

## 25.7 Assignments

1. A ski hill was designed in [2] using

$$h = 2 \cos(0.4x) \cos(0.4y) + 5xye^{-(x^2+y^2)} + 3e^{-((x-2)^2+(y-2)^2)}.$$

Choose appropriate ranges for  $x$  and  $y$  and adjust the orientation so you can clearly see three peaks.

Can you add another peak?

2. Consider the surface defined by  $z = \cos(x) \sin(y)$ , for  $x = -\pi \dots \pi$  and  $y = -\pi \dots \pi$ .

Give all Maple commands to make a 3D animation of 10 frames that gradually flattens this surface, using the formula

$$\cos((1 - 0.1k)x) \sin(1 - 0.1k)y), \quad \text{for } k = 1, 2, \dots, 10.$$

3. The *Viviani curve* is a space curve defined by  $x = R \cos^2(t)$ ,  $y = R \cos(t) \sin(t)$ , and  $z = R \sin(t)$ , where  $R$  is some parameter.

(a) Give the Maple commands to make a plot of this curve, for  $R = 1$ , using a tube around the curve of radius 0.05.

(b) Make an animation for the curve  $x = R \cos^2(t)$ ,  $y = \cos(t) \sin(t)$ , and  $z = \sin(t)$ , for  $R$  going from 1 to 10 (thus using ten frames). Give all Maple commands you use to make this animation.

4. Make an animation of a spiral, using the formulas  $x = t \cos(t)$ ,  $y = t \sin(t)$ , and  $z = t$ . Let there be 30 frames in your animation, with  $t = 1, 2, \dots, 30$ .

5. Consider  $h := t \rightarrow (x^2 + y^2 - 1)(1 - t) + (x^2 + y^2 + z^2 - 1)t$  as a deformation of the cylinder  $x^2 + y^2 - 1 = 0$  at  $t = 0$  to the sphere  $x^2 + y^2 + z^2 - 1 = 0$  at  $t = 1$ .

Use  $h$  to create an animation with 50 frames which deforms the cylinder into the sphere.

6. Consider  $v := [[0,0,0], [1,0,0], [0,1,0], [0,0,1], [1,1,0], [1,0,1], [0,1,1], [1,1,1]]$ .

(a) Use `l12 := plottools[line](v[1],v[2],thickness=3,color=red)` to define the edges between the vertices in  $v$ . The cube has 12 such edges. Store all 12 edges in the sequence `edges`.

(b) Display the `edges` with a perspective determined by `projection` equal to 0.8, no axes, and `orientation` equal to `[30,70]`.

7. Draw the Riemann surface for the  $z^{1/3}$ , defined by  $w^3 - z = 0$ , for  $w = u + Iv$ .

## References

- [1] R.M. Corless. *Essential Maple 7. An introduction for Scientific Programmers*. Springer-Verlag, 2002.
- [2] R.H. Enns and G.C. McGuire. *Computer Algebra Recipes: A Gourmet's Guide to the Mathematical Models of Science*. Springer-Verlag, 2002.
- [3] A. Heck. *Introduction to Maple*. Springer-Verlag, third edition, 2003.
- [4] G. Klimek and M. Klimek. *Discovering Curves and Surfaces with Maple*. Springer-Verlag, 1997.